Visualization for Problem 19, Section 16.3

by Russell D. Blyth - blythrd@slu.edu
and Michael K. May, S.J. - maymk@slu.edu

```
> with(plottools):
  with(plots): with(LinearAlgebra):
```

We start by plotting the planes.

```
> eqn1 := z=-6;
  eqn2 := y=0;
  eqn3 := y-x=4;
  eqn4 := 2*x+y+z=4;
  planes := implicitplot3d([eqn1, eqn2, eqn3, eqn4], x=-5..6, y=-1.
  .7, z=-7..15,
    color=[blue, red, green, yellow], style=patchnogrid,
    transparency=.5,axes=normal):
  display(planes);
```

$$eqn1 := z = -6$$

$$eqn2 := y = 0$$

$$eqn3 := y - x = 4$$

$$eqn4 := 2\,x + y + z = 4$$



Next we find the lines of intersection of the planes and plot them as well, so we see the pyramid outlined.

```
> (M1,b) := GenerateMatrix({eqn1,eqn2,eqn3},{x,y,z}):
  P1 := LinearSolve(M1,b):
  (M2,b) := GenerateMatrix({eqn1,eqn2,eqn4},{x,y,z}):
  P2 := LinearSolve(M2,b):
  (M3,b) := GenerateMatrix({eqn1,eqn3,eqn4},{x,y,z}):
  P3 := LinearSolve(M3,b):
  (M4,b) := GenerateMatrix({eqn2,eqn3,eqn4},{x,y,z}):
  P4 := LinearSolve(M4,b):
  L1 := P1*t + P2*(1-t): l1 := [L1[1],L1[2],L1[3]]:
  L2 := P1*t + P3*(1-t): l2 := [L2[1],L2[2],L2[3]]:
  L3 := P1*t + P4*(1-t): l3 := [L3[1],L3[2],L3[3]]:
  L4 := P2*t + P3*(1-t): l4 := [L4[1],L4[2],L4[3]]:
  L5 := P2*t + P4*(1-t): l5 := [L5[1],L5[2],L5[3]]:
  L6 := P3*t + P4*(1-t): l6 := [L6[1],L6[2],L6[3]]:
  lines := spacecurve({l1,l2,l3,l4,l5,l6},
     t=0..1,color=black, thickness=3):
  display([planes,lines]);
```



We also show the same plot with the faces as well:

```
> f1 := plot3d([(10-u)*t/2+(u-4)*(1-t), u, -6], t=0..1,u=0..6,
  transparency=0,color=blue,grid=[20,20],style=patchnogrid):
  f2 := plot3d([u, 0, -6*t+(1-t)*(4-2*u)], t=0..1,u=-4..5,
  transparency=0,color=red,grid=[20,20],style=patchnogrid):
  f3 := plot3d([u, u+4, -6*t+(1-t)*(-3*u)], t=0..1,u=-4..2,
```

```
transparency=0,color=green,grid=[20,20],style=patchnogrid):
f4 := plot3d([t*(4-u)/2+(1-t)*(-u)/3, 4-u-2*(t*(4-u)/2+(1-t)*(-u)
/3), u],
     t=0..1,u=-6..12, transparency=0,color=yellow,grid=[20,20],
style=patchnogrid):
display([planes,lines,f1,f2,f3,f4]);
```



From now on we plot just the parts of the planes that make the sides of the pyramid.
Rotate the plot to see all sides of the pyramid

```
> display([f1,f2,f3,f4,lines], axes=normal);
```



To set up the integral over this region, determine the slice at each height z0. (Change the value of z0 to change the location of the slice.)

```
> ft1 := plot3d([(10-u)*t/2+(u-4)*(1-t), u, -6], t=0..1,u=0..6,
  transparency=0.6,color=blue,grid=[20,20],style=patchnogrid):
  ft2 := plot3d([u, 0, -6*t+(1-t)*(4-2*u)], t=0..1,u=-4..5,
  transparency=0.6,color=red,grid=[20,20],style=patchnogrid):
  ft3 := plot3d([u, u+4, -6*t+(1-t)*(-3*u)], t=0..1,u=-4..2,
  transparency=0.6,color=green,grid=[20,20],style=patchnogrid):
  ft4 := plot3d([t*(4-u)/2+(1-t)*(-u)/3, 4-u-2*(t*(4-u)/2+(1-t)*(-
  u)/3), u],
       t=0..1,u=-6..12, transparency=0.6,color=yellow,grid=[20,20],
  style=patchnogrid):
  z0 := 1:
  sl := plot3d([(4-z0-u*(12-z0)/18)*t/2+(u*(12-z0)/18-4)*(1-t), u*
  (12-z0)/18, z0],
         t=0..1,u=0..6, transparency=0,color=black,grid=[20,20],
  style=patchnogrid):
  display([ft1,ft2,ft3,ft4,lines,sl], axes=normal);
  display([lines,sl], axes=normal);
```


12
10
8
6
4
2
0
-4
-6
-1
-2
-3
-4
1
2
3
4
5
4
5
6



Now we want to set up the triple integral. We see that z runs from -6 to the z-coordinate at which eqn2, eqn3 and eqn4 meet.

```
> solve({eqn2,eqn3,eqn4});
```

$$\{x = -4, y = 0, z = 12\}$$

So z runs from -6 to 12.

At a given height z, we note that if we integrate first in x and then in y, y runs from 0 to the intersection of eqn3, eqn4 and the plane at height z.

```
> solve({eqn3,eqn4},{x,y});
```

$$\left\{x = -\frac{1}{3}\,z, y = 4 - \frac{1}{3}\,z\right\}$$

So y runs from 0 to 4 - z/3.

Finally, for a given value of y and z, x runs from the back (green) plane to the front (yellow) plane:

```
> solve({eqn3},{x});
  solve({eqn4},{x});
```

$$\{x = y - 4\}$$

$$\left\{x = -\frac{1}{2}\,y - \frac{1}{2}\,z + 2\right\}$$

Add the range of integration for x to the plot for a fixed height z0 and value y0 of y (change y0 to anything from 0 to 4-z0/3).

```
> y0:= 1.5;
  xr:=spacecurve([t*(y0-4)+(1-t)*(2-y0/2-z0/2) ,y0,z0],t=0..1,
  color=white,thickness=3):
  display([ft1,ft2,ft3,ft4,lines,sl,xr], axes=normal);
  display([lines,sl,xr], axes=normal);
```

$$y0 := 1.5$$





So here's the integral that computes the volume of the pyramid:

```
> Int(Int(Int(1, x=y-4..2-y/2-z/2),y=0..4-z/3),z=-6..12);
```

$$\int_{-6}^{12}\int_{0}^{4-\frac{1}{3}z}\int_{y-4}^{-\frac{1}{2}y-\frac{1}{2}z+2} 1\,dx\,dy\,dz$$

Compute in steps:

```
> Int(Int(Int(1, x=y-4..2-y/2-z/2),y=0..4-z/3),z=-6..12) = Int(Int
  (int(1, x=y-4..2-y/2-z/2),y=0..4-z/3),z=-6..12);
```

$$\int_{-6}^{12}\int_{0}^{4-\frac{1}{3}z}\int_{y-4}^{-\frac{1}{2}y-\frac{1}{2}z+2} 1\,dx\,dy\,dz=\int_{-6}^{12}\int_{0}^{4-\frac{1}{3}z}\left(-\frac{3}{2}y-\frac{1}{2}z+6\right)dy\,dz$$

```
> Int(Int(Int(1, x=y-4..2-y/2-z/2),y=0..4-z/3),z=-6..12) = Int(int
  (int(1, x=y-4..2-y/2-z/2),y=0..4-z/3),z=-6..12);
```

$$\int_{-6}^{12}\int_{0}^{4-\frac{1}{3}z}\int_{y-4}^{-\frac{1}{2}y-\frac{1}{2}z+2} 1\,dx\,dy\,dz=\int_{-6}^{12}\left(-\frac{3}{4}\left(4-\frac{1}{3}z\right)^2-\frac{1}{2}z\left(4-\frac{1}{3}z\right)+24\right.$$

$$-2z\Big)\,dz$$

```
> Int(Int(Int(1, x=y-4..2-y/2-z/2),y=0..4-z/3),z=-6..12) = int(int
  (int(1, x=y-4..2-y/2-z/2),y=0..4-z/3),z=-6..12);
```

$$\int_{-6}^{12}\int_{0}^{4-\frac{1}{3}z}\int_{y-4}^{-\frac{1}{2}y-\frac{1}{2}z+2} 1\,dx\,dy\,dz = 162$$